OPERATIONS WITH SUBSPACES

In R^4 let us consider the following vector subspaces
S=<(1,-1,2,-3), (2,-2.4.-3), (1,1,2,0)>
T=<(0,-2,0,-3),(1,0,1,0)>

With respect to the standard basis of R^4, compute the cartesian equations, the parametric equations, the dimension and a basis of S, T, S+T, and S interseccion T.
Are S and T complementary subspaces?

```
> restart;
> with(linalg):
```

S

```
> v1:=[1,-1,2,-3];v2:=[2,-2,4,-3];v3:=[1,1,2,0];
```

$$v1 := [1, -1, 2, -3]$$
$$v2 := [2, -2, 4, -3]$$
$$v3 := [1, 1, 2, 0] \quad \textbf{(1.1)}$$

```
> MS:=matrix([v1,v2,v3]);
```

$$MS := \begin{bmatrix} 1 & -1 & 2 & -3 \\ 2 & -2 & 4 & -3 \\ 1 & 1 & 2 & 0 \end{bmatrix} \quad \textbf{(1.2)}$$

```
> rank(MS);
```

$$3 \quad \textbf{(1.3)}$$

dim(S)=3 and BS={v1,v2,v3} is a basis.
The parametric equations are:

```
> [x1,x2,x3,x4]=expand(a*v1+b*v2+c*v3);
```

$$[x1, x2, x3, x4] = [c + 2\,b + a, c - 2\,b - a, 2\,c + 4\,b + 2\,a, -3\,b - 3\,a] \quad \textbf{(1.4)}$$

The number of cartesian equations of S is 4-dim(S)=1

```
> ES:=matrix([[x1,x2,x3,x4],v1,v2,v3]);
```

$$ES := \begin{bmatrix} x1 & x2 & x3 & x4 \\ 1 & -1 & 2 & -3 \\ 2 & -2 & 4 & -3 \\ 1 & 1 & 2 & 0 \end{bmatrix} \quad \textbf{(1.5)}$$

```
> cartS:=det(ES)=0;
```

$$cartS := 12\,x1 - 6\,x3 = 0 \quad \textbf{(1.6)}$$

T

```
> w1:=[0,-2,0,-3];w2:=[1,0,1,0];
```

$$w1 := [0, -2, 0, -3]$$
$$w2 := [1, 0, 1, 0] \quad (2.1)$$

```
> MT:=matrix([w1,w2]);
```

$$MT := \begin{bmatrix} 0 & -2 & 0 & -3 \\ 1 & 0 & 1 & 0 \end{bmatrix} \quad (2.2)$$

```
> rank(MT);
```

$$2 \quad (2.3)$$

dim(T)=2 and BT={w1,w2} is a basis.
The parametric equations are:

```
> [x1,x2,x3,x4]=expand(a*w1+b*w2);
```

$$[x1, x2, x3, x4] = [b, -2\,a, b, -3\,a] \quad (2.4)$$

The number of cartesian equations of T is 4-dim(T)=2

```
> ET:=matrix([[x1,x2,x3,x4],w1,w2]);
```

$$ET := \begin{bmatrix} x1 & x2 & x3 & x4 \\ 0 & -2 & 0 & -3 \\ 1 & 0 & 1 & 0 \end{bmatrix} \quad (2.5)$$

```
> ET1:=submatrix(ET,1..3,1..3);ET2:=submatrix(ET,1..3,[1,2,4]);
```

$$ET1 := \begin{bmatrix} x1 & x2 & x3 \\ 0 & -2 & 0 \\ 1 & 0 & 1 \end{bmatrix}$$

$$ET2 := \begin{bmatrix} x1 & x2 & x4 \\ 0 & -2 & -3 \\ 1 & 0 & 0 \end{bmatrix} \quad (2.6)$$

```
> cartT1:=det(ET1);cartT2:=det(ET2);
```

$$cartT1 := -2\,x1 + 2\,x3$$
$$cartT2 := -3\,x2 + 2\,x4 \quad (2.7)$$

S+T

S+T=<v1,v2,v3,w1,w2>

```
> MST:=stackmatrix(MS,MT);
```

(3.1)

$$MST := \begin{bmatrix} 1 & -1 & 2 & -3 \\ 2 & -2 & 4 & -3 \\ 1 & 1 & 2 & 0 \\ 0 & -2 & 0 & -3 \\ 1 & 0 & 1 & 0 \end{bmatrix} \quad (3.1)$$

```
> rank(MST);
```

$$4 \quad (3.2)$$

Thus dim(S+T)=4 and S+T=R^4 and it has no equations.

S∩T

By the Grassman formula dim(S∩T)= dim(S)+dim(T)-dim(S+T)=3+2-4=1 and the number of cartesian equations is 3.

```
> cartS;cartT1;cartT2;
```

$$12\,x1 - 6\,x3 = 0$$
$$-2\,x1 + 2\,x3$$
$$-3\,x2 + 2\,x4 \quad (4.1)$$

```
> solve({cartS,cartT1,cartT2},{x1,x2,x3,x4});
```

$$\left\{x1 = 0,\ x2 = x2,\ x3 = 0,\ x4 = \frac{3}{2}\,x2\right\} \quad (4.2)$$

```
> parSintT:=[0,a,0,3*a/2];
```

$$parSintT := \left[0,\ a,\ 0,\ \frac{3}{2}\,a\right] \quad (4.3)$$

A basis of S∩T is {(0,1,0,3/2)}.

S and T are not complementary because their intersection is not zero.

NEW COMMANDS USED:

stackmatrix(A,B,...) joints several matrices vertically
submatrix(A, Rrange, Crange) submatrix of A with rows in Rrange and columns in Crange.

Exercice 20, Sheet 1.